モエッスナーの魔法の数

SHIMURA Masato

2016 年 10 月 20 日

目次

コンウエイとガイの名著「数の本」にモエッスナーの魔法の数が取り上げられている。

1　三角数, 平方数と立方数

三角数はピタゴラスの時代から熱心に研究され、近世ではフェルマ、ルジャンドル、ガウスなども熱心に手を染めている。

三角数を求める



n 番目の三角数を Δ_n であらわす。

次の図から三角数 Δ_n の 2 倍は連続する 2 個の整数の積 $n(n+1)$ になっているので、n 番目の三角数 $\Delta_n = \dfrac{1}{2}n(n+1)$ となる。





次の等式が成り立っているとする

$$1 + 2 + \cdots + (n-1) = \frac{1}{2}(n-1)n = \frac{1}{2}n^2 - \frac{1}{2}n$$

これに n を足すと目的の式になる

$$\left(\frac{1}{2}n^2 - \frac{1}{2}n\right) + n = \frac{1}{2}n^2 + \frac{1}{2}n = \frac{1}{2}n(n+1)$$

右側の図は整然とリードが並んだパイプオルガンに見立てて「パイプオルガン法」というようだ。ここから等差数列の式が現れる

等間隔に並ぶ n 個の数の最初を a、最後を z とすると、その n 個の数の総和は、その平均値の n 倍である	$n \times \frac{a+z}{2}$

三角数は J の簡潔なスクリプトで求めることができる

```
   <\>: i. 6
+-+---+-----+-------+---------+-----------+
|1|1 2|1 2 3|1 2 3 4|1 2 3 4 5|1 2 3 4 5 6|
+-+---+-----+-------+---------+-----------+
   +/\>: i. 6
1 3 6 10 15 21
```

J Grammar

1. `>: i.6` は *1* から始まる *6* 個の数列を生成する。`>:`は *C* の `++`
2. `<`は *Box* に入れる
3. `\` は *Prefix* で、逐次 *1* 項ずつ増加してボックスで囲む
4. `+/`は $\sum$

5. +/\と *Box* < の代わりに + を用いると<ボックスに入れ、ボックス内で演算してから開く一連の操作の記述を省略できる

1.1 3 角数と平方数

1 から始まる連続する奇数を一個、二個、三個、四個と足していくと平方数になる。
それを重複しないで続きから始めると立方数になる。

$$\begin{array}{rrrrrl|}
 & & & & 1 & = 1^2 \\
 & & & 1 & +3 & = 2^2 \\
 & & 1 & +3 & +5 & = 3^2 \\
 & 1 & +3 & +5 & +7 & = 4^3 \\
1 & +3 & +5 & +7 & +9 & = 5^2
\end{array}
\qquad
\begin{array}{rrrrrl|}
 & & 1 & & & = 1^3 \\
 & 3 & +5 & & & = 2^3 \\
 & 7 & +9 & +11 & & = 3^3 \\
13 & +15 & +17 & +19 & & = 4^3 \\
21 & +23 & +25 & +27 & +29 & = 5^3
\end{array}$$

連続する *3* 角数の和は平方数になる




$$\begin{aligned}
\Delta_n + \Delta_{n-1} &= 1 + 2 + 3 + \cdots + n \\
&\qquad +1 + 2 + \cdots + (n-1) \\
&= 1 + 3 + 5 + \cdots + (2n-1)
\end{aligned}$$

平方数を求める *J* のスクリプト。力業で三角数を *2* 個くっつけて重複する対角部分を抜いている

```
   ] a=. <\>: i. 6
+-+---+-----+-------+---------+-----------+
|1|1 2|1 2 3|1 2 3 4|1 2 3 4 5|1 2 3 4 5 6|
+-+---+-----+-------+---------+-----------+
   ; +/ L:0  a, L:0  |.@  }: L:0 a
1 4 9 16 25 36
```

J Grammar

1. }: 末尾を落とす
2. *L:0* ボックスの状態のままでボックスごとに計算する

3. **|.** 回転させる。デフォルトでは左右の反転
4. **;** ボックスを開く

2 モエッスナーの魔法の数

2.1 平方数

1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
1		4		9		16		25		36		49		64	

アルゴリズム .

1. 偶数に印をつける
   [*1]
2. 残った奇数（黒）を逐次足し上げると平方数になる

J の Script　モエッスナーのアルゴリズムに忠実に *J* のスクリプトを作成する。

- *2* 個ずつに区切り、ボックスに入れる。 _2 は非重複で *2* 個づつ区切る

```
    _2<\>:i.16
+---+---+---+---+----+-----+-----+-----+
|1 2|3 4|5 6|7 8|9 10|11 12|13 14|15 16|
+---+---+---+---+----+-----+-----+-----+
```

- ボックスの中の後尾を落とし、ボックスを開く。

```
   ; }: (L:0) _2<\>:i.16
1 3 5 7 9 11 13 15
```

- 足し上げる

```
   +/\   ; }: (L:0) _2<\>:i.16
1 4 9 16 25 36 49 64
```

Script　*Explicit(*明示型*)* で組上げる。

```
box_in=: 4 : ' (-x) <\ >:i. y'
droplast=: 4 : ' ; }: L:0 x box_in y'
onebyone=: 4 : ' +/\ x droplast y'

moessner2=: 4 : ' (x box_in y),: {@> x onebyone y'
```

[*1] 原著では ○ で囲んでいるが TeX の丸囲み記号の有無が分からないので赤文字にした

```
   2 moessner2 16
+---+---+---+---+----+-----+-----+-----+
|1 2|3 4|5 6|7 8|9 10|11 12|13 14|15 16|
+---+---+---+---+----+-----+-----+-----+
|1  |4  |9  |16 |25  |36   |49   |64   |
+---+---+---+---+----+-----+-----+-----+
```

平方数と立方数　　奇数を足すと平方数や立方数になる

2.2　立方数

(3 個づつに区切り）*3* 番目の数に印をつけて残りの数を加えていき、各ブロックの最後の数に印をつける。そして、印のない数を加えると立方数になる。

1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
1	3		7	12		19	27		37	48		61	75	
1			8			27			64			125		

```
   ^&3 >:i.5
1 8 27 64 125
```

2.3　4 乗数

1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
1	3	6		11	17	24		33	43	54		67	81	96	
1	4			15	32			65	108			175	256		
1				16				81				256			
1^4				2^4				3^4				4^4			

```
   ^&4 >:i.5
1 16 81 256 625
```

2.4　3 角数は階乗数

1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
	2		6	11		18	26	35		46	58	71	85	
			6			24	50			96	154	225		
						24				120	274			
										120				

```
   !>:i.5
1 2 6 24 120
```

2.5　4 角数

1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	15
	2	5		10	16	23	31		41	52	64	77	91	106	
	2			12	28	51			92	144	208	285	376		
				12	40				132	276	484	769			
				12					144	420	904				
									144	564					
									144						

平方数を求める

$$
\begin{array}{c|ccccccc}
1 & & & & 1 & & & \\
2 & & & 1 & +2 & +1 & & \\
3 & & 1 & +2 & +3 & +2 & +1 & \\
4 & 1 & +2 & +3 & +4 & +3 & +2 & +1
\end{array}
$$

平方数の符号を × にすると最後の数が現れる。

$$
\begin{array}{ccccccc}
 & & & 1 & & & = 1 \\
 & & 1 & \times 2 & \times 1 & & = 2 \\
 & 1 & \times 2 & \times 3 & \times 2 & \times 1 & = 12 \\
1 & \times 2 & \times 3 & \times 4 & \times 3 & \times 2 & \times 1 = 144
\end{array}
$$

3　多角数

多角数はピタゴラスの時代から熱心に取り上げられていた。

n 番目の p 角数とすると

$$\frac{(p-2)n^2-(p-4)n}{2}$$

三角数

$$1+2+3+4+5+\cdots \rightarrow 1,3,6,10,15\cdots \rightarrow \frac{1}{2}(n^2+n)$$

平方数

$$1+3+5+7+11+\cdots \rightarrow 1,4,9,16,25\cdots \rightarrow \frac{1}{2}(2n^2)$$

五角数

$$1+4+7+10+13+\cdots \rightarrow 1,5,12,22,35\cdots \rightarrow \frac{1}{2}(3n^2-n)$$

六角数

$$1+5+9+13+17+\cdots \rightarrow 1,6,15,28,45\cdots \rightarrow \frac{1}{2}(4n^2-2n)$$

七角数

$$1+6+11+16+21+\cdots \rightarrow 1,7,18,34,55\cdots \rightarrow \frac{1}{2}(5n^2-3n)$$

八角数

$$1+7+13+19+25+\cdots \rightarrow 1,8,21,40,65\cdots \rightarrow \frac{1}{2}(6n^2-4n)$$

多角数のもとになる数列の打出しは、*Tacit* のループを使うと簡単である。

三角数 `+/\>:i.10`

```
    1 3 6 10 15 21 28 36 45 55
```

平方数 .

```
      +&2 ^:(i.10) 1
   1 3 5 7 9 11 13 15 17 19
     +/\ +&2 ^:(i.10) 1
   1 4 9 16 25 36 49 64 81 100
```

五角数 *3* 間隔の等差数列

```
      +&3 ^:(i.10) 1
   1 4 7 10 13 16 19 22 25 28
    +/\  +&3 ^:(i.10) 1
   1 5 12 22 35 51 70 92 117 145
```

六角数 *4* 間隔の等差数列

```
   +/\ +&4 ^:(i.10) 1
1 5 9 13 17 21 25 29 33 37
    +/\ +&4 ^:(i.10) 1
1 6 15 28 45 66 91 120 153 190
```

七角数　*5* 間隔の等差数列

```
    +&5 ^:(i.10) 1
1 6 11 16 21 26 31 36 41 46
   +/\ +&5 ^:(i.10) 1
1 7 18 34 55 81 112 148 189 235
```

八角数　*6* 間隔の等差数列

```
    +&6 ^:(i.10) 1
1 7 13 19 25 31 37 43 49 55
  +/\  +&6 ^:(i.10) 1
1 8 21 40 65 96 133 176 225 280
```

汎用関数　多角数を求める汎用関数のスクリプト

```
    5 multiangler >:i.5
1 5 12 22 35
```

```
 multiangler=:4 : '-:(-(x-4)*y)  +(x-2)*y^2'
```

J Grammar　.

1. `+/\ +&2 ^:(i.10) 1`
2. `^:(i.10)`　*Tacit* のループで *10* 回反復
3. `+&2`　+2

4　多角数は数論の宝庫

4.1　完全数

完全数は約数の和がそれ自身となる数である。
完全数はユークリッドの原論に証明が残されている

$$2^{p-1}(2^p-1)$$

```
  perfect 2 3 4 5  6 7
  x: perfect 2 3  5  7 13 17 19
6 28 496 8128 33550336 8589869056 137438691328
```

$1 + 2 + 4 + 7 + 14 = 28$
完全数は三角数である

```
    ({@> perfect 2 3 5 7 11 17) e. L:0 +/\i.1000000
+-+-+-+-+-+-+
|1|1|1|1|1|1|
+-+-+-+-+-+-+
```

J Grammar

1. *Script* `perfect=: 3 : '(2^<:y)*<:2^y '`
2. 素因数分解は *q:*

```
   q: 1072
2 2 2 2 67
```

3. 約数を素因数分解から作成するのは意外に厄介だ。

```
   yakusuu 1072
1 2 4 8 16 67 134 268 536
```

素因数分解の数を反転させたものを前にくっつけて累加して重複を整理する。

`yakusuu=: 3 : '}: 1, ~. /:~ */\ |. q: y), */\ q: y'`

$p = 11$ は完全数ではない

```
   x: perfect 2 3 5 7 11 13 17 19
6 28 496 8128 2096128 33550336 8589869056 137438691328
   ; +/ L:0 yakusuu L:0 {@> perfect 2 3 5 7  11 13 17 19
6 28 496 8128 2119769 33550336 8589869056 137438691328
```

4.2 三角数でかつ平方数}

N 番目の三角数が M 番目の平方数になる場合

$$\frac{1}{2}N(N+1) = M^2$$

```
   +/\>:i.10
1 3 6 10 15 21 28 36 45 55
   (;({@>+/\>:i.10)e.L:0+/\+&2^:(i.10)1)
1 0 0 0 0 0 0 1 0 0
```

```
   (;({@>+/\>:i.1000)e.L:0+/\+&2^:(i.1000)1)    #+/\>:i.1000
1 36 1225 41616
```

$x = ”N + 1, y = 2M$ とおくと、ペル方程式が現れる

$$x^2 - 2y^2 = 1$$

J Grammar

1. *# Copy* 指標 *1* の個所を *Copy* して取り出す

1. ***e.** Members of* x が y に含まれているか否かを走査する。
2. `{@>` `L:0` 完全数をボックスを用いて一個づづ個別演算する。ボックスを用いないと *6 28..* と組み合わさった数として取り扱われる

4.3 パスカルの三角形

- パスカルの三角形の *3* 列目は三角数になっている
- *4* 列目は四面体数になる。
- 右下段から袈裟掛けにするとフィボナッチ数が姿を現す

$$\begin{aligned}
1 &= 1 \\
3 &= 1 \quad 2 \\
5 &= 1 \quad 3 \quad 1 \\
8 &= 1 \quad 4 \quad 3 \\
13 &= 1 \quad 5 \quad 6 \quad 1 \\
21 &= 1 \quad 6 \quad 10 \quad 4
\end{aligned}$$

$_nC_r$

```
   |: (i.7)!/2 3 4 5 6
1 2  1  0  0 0 0
1 3  3  1  0 0 0
1 4  6  4  1 0 0
1 5 10 10  5 1 0
1 6 15 20 15 6 1
```

$$S_n = \frac{1}{2}n(n+1)$$

```
   2!>:i.10   NB. nC3
0 1 3 6 10 15 21 28 36 45

   +/\>:i.10  NB. 三角数
1 3 6 10 15 21 28 36 45 55
```

$$ {}_nP_r = \frac{n!}{(n-r)!} $$

```
   (! 10)% !7
720

   (!10)% (!7)*(!3)
120
   7!10
120
```

$$ \binom{n}{r} = {}_nC_r = \frac{{}_nP_r}{r!} = \frac{n!}{r!(n-r)!} $$

$$ \binom{n}{k} = \frac{P(n,k)}{k!} $$

5 Script

```
NB. Issue: Conway Gui The Book of Numbers Chap 3
NB. Moessner Numbers

box_in=: 4 : ' (-x) <\ >:i. y'
droplast=: 4 : ' ; }: L:0 x box_in y'
onebyone=: 4 : ' +/\ x droplast y'
moessner2=: 4 : ' (x box_in y),: {@> x onebyone y'


perfect=: 3 : '(2^<:y)*<:2^y '
```

References

J.H. コンウエイ ·*R.K.* ガイ 根上生也訳「数の本」　*Springer-Verlag* 東京　*2001*

J 言語は次から *DL* できます。*WIN* 版ならサイズの大きい *QT* 版が便利です。インストールしたら *Tool* → *PackageManager* でパッケージを *(*全部*)* 導入してください。

jsoftware.com → *Download*

執筆時点での最新版は *J805*